

## NAG Library Routine Document

### G01EAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

G01EAF returns a one or two tail probability for the standard Normal distribution, via the routine name.

#### 2 Specification

```
FUNCTION G01EAF (TAIL, X, IFAIL)
REAL (KIND=nag_wp) G01EAF
INTEGER          IFAIL
REAL (KIND=nag_wp) X
CHARACTER(1)    TAIL
```

#### 3 Description

The lower tail probability for the standard Normal distribution,  $P(X \leq x)$  is defined by:

$$P(X \leq x) = \int_{-\infty}^x Z(X) dX,$$

where

$$Z(X) = \frac{1}{\sqrt{2\pi}} e^{-X^2/2}, -\infty < X < \infty.$$

The relationship

$$P(X \leq x) = \frac{1}{2} \operatorname{erfc}\left(\frac{-x}{\sqrt{2}}\right)$$

is used, where  $\operatorname{erfc}$  is the complementary error function, and is computed using S15ADF. For the upper tail probability the relationship  $P(X \geq x) = P(X \leq -x)$  is used and for the two tail significance level probability twice the probability obtained from the absolute value of  $x$  is returned.

When the two tail confidence probability is required the relationship

$$P(X \leq |x|) - P(X \leq -|x|) = \operatorname{erf}\left(\frac{|x|}{\sqrt{2}}\right),$$

is used, where  $\operatorname{erf}$  is the error function, and is computed using S15AEF.

#### 4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

## 5 Arguments

1: TAIL – CHARACTER(1) *Input*

*On entry:* indicates which tail the returned probability should represent.

TAIL = 'L'

The lower tail probability is returned, i.e.,  $P(X \leq x)$ .

TAIL = 'U'

The upper tail probability is returned, i.e.,  $P(X \geq x)$ .

TAIL = 'S'

The two tail (significance level) probability is returned, i.e.,  $P(X \geq |x|) + P(X \leq -|x|)$ .

TAIL = 'C'

The two tail (confidence interval) probability is returned, i.e.,  $P(X \leq |x|) - P(X \leq -|x|)$ .

*Constraint:* TAIL = 'L', 'U', 'S' or 'C'.

2: X – REAL (KIND=nag\_wp) *Input*

*On entry:*  $x$ , the value of the standard Normal variate.

3: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If IFAIL  $\neq$  0, then G01EAF returns 0.0.

IFAIL = 1

On entry, TAIL  $\neq$  'L', 'U', 'S' or 'C'.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

Accuracy is limited by *machine precision*. For detailed error analysis see S15ADF and S15AEF.

## 8 Parallelism and Performance

G01EAF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

Four values of TAIL and X are input and the probabilities calculated and printed.

### 10.1 Program Text

```

Program g01eafe
!      G01EAF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
Use nag_library, Only: g01eaf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: prob, x
Integer                    :: ifail
Character (1)              :: tail
!      .. Executable Statements ..
Write (nout,*) 'G01EAF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Display title
Write (nout,*) '  TAIL      X      Probability'
Write (nout,*)

d_lp: Do
  Read (nin,*,Iostat=ifail) tail, x
  If (ifail/=0) Then
    Exit d_lp
  End If

!      Calculate probability
  ifail = 0
  prob = g01eaf(tail,x,ifail)

!      Display results
  Write (nout,99999) tail, x, prob
End Do d_lp

99999 Format (3X,A1,4X,F5.2,6X,F7.4)
End Program g01eafe

```

## 10.2 Program Data

```
G01EAF Example Program Data
'L' 1.96 : TAIL X
'U' 1.96
'C' 1.96
'S' 1.96
```

## 10.3 Program Results

G01EAF Example Program Results

TAIL	X	Probability
L	1.96	0.9750
U	1.96	0.0250
C	1.96	0.9500
S	1.96	0.0500

---